

# OAK RIDGE NATIONAL LABORATORY

operated by

UNION CARBIDE CORPORATION

NUCLEAR DIVISION

for the

U.S. ATOMIC ENERGY COMMISSION



ORNL - TM - 1805

COPY NO. ~~1~~12

DATE - March 22, 1967

## "PUT-5" AND "TAKE-5" — TWO MAGNETIC TAPE SYSTEMS

ROUTINES FOR THE PDP-8

B. W. Rust<sup>†</sup> and W. R. Burrus

### ABSTRACT

Two magnetic tape systems routines have been written to provide fast input of programs or data to a PDP-8 computer from a DEC type 580 tape transport. PUT-5 is used to create systems tapes containing the user's programs and TAKE-5 is used to read the programs into the computer from such systems tapes.

This Work Supported by  
NATIONAL AERONAUTICS AND SPACE ADMINISTRATION  
Under Order R-104(1)

<sup>†</sup>Computer Technology Center, Oak Ridge Gaseous Diffusion Plant.

**NOTICE** This document contains information of a preliminary nature and was prepared primarily for internal use at the Oak Ridge National Laboratory. It is subject to revision or correction and therefore does not represent a final report.

N67-31743  
(THRU) 0  
(CODE) 08  
(CATEGORY)  
(PAGES)  
(ACCESSION NUMBER)  
(NASA CR OR TMX OR AD NUMBER)  
FACILITY FORM 602

#### LEGAL NOTICE

This report was prepared as an account of Government sponsored work. Neither the United States, nor the Commission, nor any person acting on behalf of the Commission:

- A. Makes any warranty or representation, expressed or implied, with respect to the accuracy, completeness, or usefulness of the information contained in this report, or that the use of any information, apparatus, method, or process disclosed in this report may not infringe privately owned rights; or
- B. Assumes any liabilities with respect to the use of, or for damages resulting from the use of any information, apparatus, method, or process disclosed in this report.

As used in the above, "person acting on behalf of the Commission" includes any employee or contractor of the Commission, or employee of such contractor, to the extent that such employee or contractor of the Commission, or employee of such contractor prepares, disseminates, or provides access to, any information pursuant to his employment or contract with the Commission, or his employment with such contractor.

PUT-5 and TAKE-5 are magnetic tape systems routines for use on a PDP-8 computer with a DEC Type 580 tape transport. Their purpose is to provide a fast means for loading programs into core. Each of them occupies all of the last page of core up to the CRIMP loader, leaving all of the first 31 pages for the programs to be loaded. Thus they occupy essentially the same space as the binary paper tape loader. PUT-5 is used to put the programs or data on tape and TAKE-5 is used to read them into core. Each time PUT-5 is used it "puts" a file on the systems tape. A file consists of three records:

1. an end of file,
2. a short ID record,
3. a long core image record.

The format of these records are such that they can be read with an IBM-7090 computer so that any systems tape made up with PUT-5 can easily be reproduced on the 7090. The end of file at the beginning of each file serves as a physical separation between the files. It is needed so that the act of "putting" a file on the tape does not erase part of the next file. The ID record is  $11_8$  words long and contains the file number, the lower and upper limits for loading at "take" time, indexing information to be used when "taking" and enough dummy words to make the record equivalent to three 7090 words. The core image record consists of  $7601_8$  PDP-8 words. The first is a dummy word which is needed to make the total number of words in the record an even multiple of 3 and hence an even number of 7090 words. The other  $7600_8$  words are the contents of the first 31 pages of core. All the files written by PUT-5 have this same format and the same length and so any file on the systems tape can be rewritten without altering any of the other files.

TAKE-5 is designed to read systems tapes of the type just described. Although the core image record always contains the contents of locations 0--7577 at "put" time, TAKE-5 does not necessarily have to load all these locations from tape at "take" time. Instead, the lower and upper limits for loading must be specified at "put" time. These are then written in the ID record which will be read before the core image record at "take" time. Although it would be more desirable to be able to specify the loading limits at "take" time there was not enough space available on the last page to do this in TAKE-5. TAKE-5 does, however, halt after reading the ID record displaying the lower and upper limits it has received from the ID record in the AC and MQ respectively. The user may then change these loading limits if he very carefully follows the instructions for doing this which are given in the section on operating instructions.

These programs have been used successfully for several months to make and read systems tapes, and although they have performed successfully most of the time they occasionally have "blown up". On at least one occasion this was found to be due to flaws in the hardware. Although we think that the programs are free of logical flaws the user should be careful, especially since PUT-5 does not check itself.

## OPERATING INSTRUCTIONS FOR PUT-5 AND TAKE-5

PUT-5: Put-5 loads into the last page from 7600 up to the CRIMP loader. (7600-7755)

1. Make sure it is loaded with the CRIMP loader.
2. Then start at 7600. Computer will halt, tape will rewind.
3. Wait till rewind complete.
4. Put file number in keys, and hit continue. Computer will halt again.
5. Put first address to be "put" in keys and hit continue. Computer will halt again.
6. Put last address to be "put" in keys and hit continue.  
(Do not "put" outside the range from 0 to 7577.)  
Then computer will "put" on tape and rewind. To put another file repeat the process beginning with Step 3.

TAKE-5: Take-5 loads into the last page from 7600 up to the CRIMP loader. (7600-7755)

1. Make sure it is loaded with the CRIMP loader.
2. Start at 7600. Tape will rewind, and computer will halt.
3. Wait until rewind complete.
4. Put file number in keys and continue.  
Computer will try to "take" the ID record and will stop with either an error code (2222 or 7777) in AC, or will stop with the first address in AC and last address in MQ. A stop with 2222 in the AC means parity error and 7777 means size error.\*
5. If there was an error, hit continue to rewind and try again, starting with Step 3.  
If there was not error, hit continue and the core image record will be "taken".
6. If successful, tape will rewind and halt. To read another file repeat the process beginning at Step 3.  
If not successful, computer will halt with an error code (2222 or 7777) in the AC. Hit continue to rewind and try again starting with Step 3.

\*Note: If the user desires to change the loading limits he must interrupt the process after Step 4. At this point he must deposit the address of the new lower limit for loading in location  $7746_8$ , the one's complement of the lower limit in location  $7754_8$ , and the two's complement of the number of words he desires to load in location  $7755_8$ . He must then restart the program at location  $7643_8$  and the program will then read the core image record loading between the new limits which he has specified.

ORNL DWG. 66-3806




Fig. 1. Flow Diagram of PUT-5.

ORNL DWG. 66-3807





Fig. 2. Flow Diagrams of Subroutines Called by PUT-5.




Fig. 3. Flow Diagram of TAKE-5.






Fig. 4. Flow Diagram of Subroutines Called TAKE-5.

/PUT5 -- THE FIFTH INTERIM MAGNETIC TAPE SYSTEM  
 / WRITES A FILE ON TAPE IN FILE POSITION GIVEN BY KEY SWITCHES  
 / THE FILE CONSISTS OF 1. AN EOF MARK  
 / 2. A SHORT ID RECORD  
 / 3. A LONG CORE IMAGE RECORD  
 / NOTE. THE ID RECORD CONTAINS THE LOADING LIMITS FOR  
 / THE CORE IMAGE RECORD.

7600	4302	*7600	JMS	READY	/REWIND THE TAPE
7601	1327	PUT5.	TAD	C0002	
7602	6707		TIFM		
7603	4302		JMS	READY	
7604	7602		HLT	CLA	/STOP WITH 0000 IN AC
7605	7604		LAS		/ AND GET FILE NO. FROM SWITCHES
7606	3345		DCA	IDENQ	
7607	7402		HLT		
7610	7604		LAS		/PICK UP FIRST ADDR. FOR CORE IMAGE REC. TO LOAD
7611	3345		DCA	ADDR1	/ INTO WHEN READ BY TAKES.
7612	7402		HLT		/ THEN STOP TO PICK UP
7613	7604		LAS		/ LAST ADDRESS TO LOAD INTO
7614	3347		DCA	ADDR2	
7615	1346		TAD	ADDR1	
7616	7040		CMA		/TAKES REQUIRES ONES COMP. OF ADDR1 TO BE READ
7617	3354		DCA	LDPRST	/ IN FROM ID RECORD
7620	1347		TAD	ADDR2	/TAKES REQUIRES TWOS COMP. OF NO. WORDS TO
7621	7040		CMA		/ BE LOADED,
7622	1346		TAD	ADDR1	
7623	1341		TAD	M1	/ ALSO TAKES WILL ADD 1 TO THIS NO. WHEN IT
7624	3355		DCA	LDLAST	/ READS THE ID. REC. SO MUST SUBTRACT 1 NOW.
7625	1345		TAD	IDENQ	/MULT. NO. FILES BY 3 AND SUBTRACT 3 TO
7626	7104		CLL	RAL	/ GET NO. RECS. TO SKIP
7627	1345		TAD	IDENQ	
7630	1340		TAD	M3	
7631	7450		SNA		
7632	5244		JMP	EOF	/IF 1ST FILE IS THE ONE TO BE WRITTEN
7633	7041		CIA		
7634	3342		DCA	COUNT	/-NO. RECORDS TO BE SKIPPED
7635	1330	SPACE,	TAD	C0530	
7636	6707		TIFM		/SKIP THE REQUIRED NO. OF RECS.
7637	6732		TCPI		
7640	5237		JMP	,=1	
7641	2342		ISZ	COUNT	
7642	5237		JMP	,=3	
7643	6724		TSST		
7644	4302	EOF,	JMS	READY	/WRITE THE EOF
7645	1336		TAD	M1661	
7646	3342		DCA	COUNT	/1661 EVEN DENSITY 0 S IN EOF SPACE
7647	1331		TAD	C2630	
7650	6707		TIFM		
7651	7200		CLA		
7652	6716	SPCLP,	TSWR		/ WRITE EVEN PARITY ZEROS
7653	6721		TSDF		
7654	5253		JMP	,=1	
7655	2342		ISZ	COUNT	
7656	5252		JMP	SPCLP	
7657	1337		TAD	EOFC	
7660	6716		TSWR		/ WRITE THE EOF MARK
7661	6721		TSDF		

7662	5261	JMP	,-1	
7663	6724	TSST		
7664	4302	JMS	READY	/NOW WRITE ID RECORD
7665	1333	TAD	IDLOC	
7666	3343	DCA	ADDR	
7667	1334	TAD	M11	
7670	3342	DCA	COUNT	
7671	4311	JMS	WRITE	
7672	4302	JMS	READY	/NOW WRITE CORE RECORD
<hr/>				
7673	7240	CLA CMA		/FIRST WRITE WORD 7777, THEN WORDS 0000-7577
7674	3343	DCA	ADDR	
7675	1335	TAD	M7601	
7676	3342	DCA	COUNT	
7677	4311	JMS	WRITE	
7700	4302	JMS	READY	
7701	5200	JMP	PUT5	/JUMP TO NORMAL STOP
7702	0	READY,	0	
7703	7200	CLA		
7704	6734	TSRS		/ACCUMULATOR = STATUS
7705	7006	RTL		
7706	7700	SMA CLA		
7707	5304	JMP	READY+2	
7710	5702	JMP I	READY	
7711	0	WRITE,	0	
7712	1332	TAD	C2730	
7713	6707	TIFM		
7714	7200	WLOOP,	CLA	
7715	1743	TAD I	ADDR	
7716	6716	TSWR		
7717	6721	TSDE		
7720	5317	JMP	,-1	
7721	2343	ISZ	ADDR	
7722	7000	OPR		
7723	2342	ISZ	COUNT	
7724	5314	JMP	WLOOP	
7725	6724	TSST		
7726	5711	JMP I	WRITE	
7727	2	C0002,	0002	
7730	530	C0530,	0530	
7731	2630	C2630,	2630	
7732	2730	C2730,	2730	
7733	7745	IDLOC,	7745	
7734	7767	M11,	7767	
7735	177	M7601,	0177	
7736	6117	M1661,	6117	
7737	17	EMFC,	17	
7740	7775	M3,	7775	
7741	7777	M1,	7777	
7742	0	COUNT,	0	
7743	0	ADDR,	0	
		/		
		*7745		
7745	0	IDENC,	0	
7746	0	ADDR1,	0	
7747	0	ADDR2,	0	
7750	7000	OPR		
7751	7000	OPR		
7752	7000	OPR		
7753	0	IDENC,	0	
7754	0	LDPRST,	0	

7755 0 LDLAST, 0  
/

---

SYMBOL TABLE

ADDR	7743
ADDR1	7746
ADDR2	7747
C0002	7727
C0530	7730
C2630	7731
C2730	7732
COUNT	7742
EOF	7644
EOFC	7737
IDENO	7745
IDEN6	7753
IDLAC	7733
LDIRST	7754
LDLAST	7755
M1	7741
M11	7734
M1661	7736
M3	7740
M7601	7735
PUT5	7600
READY	7702
SPACE	7635
SPCLP	7652
WLOOP	7714
WRITE	7711

---

DUPLICATE TAGS

NONE

UNDEFINED SYMBOLS

NONE

---

## /TAKE5 -- THE 5TH INTERIM MAGNETIC TAPE SYSTEM

		/		READS THE ID RECORD AND DATA RECORD FROM FILE POSITION	
		/		GIVEN BY KEY SWITCHES	
		/			
7600	4253	JMS	READY		
7601	1331	TAD	C0002	/REWIND THE TAPE	
7602	6707	TIFM			
7603	4253	JMS	READY		
7604	7402	HLT			
7605	7604	LAS		/PICK UP FILE NO.	
7606	3337	DCA	COUNT		
7607	1337	TAD	COUNT	/MULTIPLY IT BY 3 AND SUBTRACT 2 TO GET	
7610	7104	CLL RAL		/ NO. OF RECORDS TO SKIP	
7611	1337	TAD	COUNT		
7612	1342	TAD	M2		
7613	7041	CIA			
7614	3337	DCA	COUNT		
7615	1332	TAD	C0530	/SPACE OVER TO DESIRED ID RECORD	
7616	6707	TIFM			
7617	6732	ICPI			
7620	5217	JMP	,=1		
7621	2337	ISZ	COUNT		
7622	5217	JMP	,=3		
7623	6724	TSST			
7624	4253	JMS	READY		
7625	1343	TAD	IDLOC	/READ IN THE ID RECORD -- 9 WORDS IN ALL	
7626	3340	DCA	ADDR		
7627	1334	TAD	OC11		
7630	3337	DCA	COUNT		
7631	7040	CMA			
7632	3354	DCA	LDFRST		
7633	1341	TAD	M11		
7634	3355	DCA	LDLAST		
7635	4262	JMS	READ		
7636	4253	JMS	READY		
7637	1347	TAD	ADDR2		
7640	7421	MOL			
7641	1346	TAD	ADDR1		
7642	7402	HLT		/STOP DISPLAYING ADDRESSES OF FIRST AND LAST	
				/ LOCATIONS TO LOAD DATA INTO FROM CORE IMAGE	
				/ RECORD.	
7643	7200	CLA		/CHANGE LOAD ADDRESSES AT YOUR OWN RISK. TO DO	
7644	1346	TAD	ADDR1	/ SO YOU MUST CHANGE ADDR1, LDFRST AND LDLAST	
7645	3340	DCA	ADDR	/HIT CONTINUE TO READ CORE IMAGE RECORD	
7646	1335	TAD	OC7601		
7647	3337	DCA	COUNT		
7650	4262	JMS	READ		
7651	4253	JMS	READY		
7652	5200	JMP	START		
7653	0	READY,	0000	/SUBROUTINE FOR WAITING UNTIL TAPE READY	
7654	7200	CLA			
7655	6734	TSRS		/STATUS TO AC	
7656	7006	RTL			
7657	7700	SMA CLA		/SKIP IF TRANSPORT READY	

7660	5255		JMP	,*3	
7661	5653		JMP	I	READY / RETURNS WITH AC CLEARED
/					
7662	0	READ,	0000		/SUBROUTINE TO READ RECORD FROM TAPE.
7663	7100		CLL		/ ADDR SHOULD CONTAIN ADDR. OF 1ST LOC. TO
7664	1333		TAD	C2570	/ ACTUALLY STORE DATA INTO,
7665	6707		TIFM		/LDFRST MUST CONTAIN ONES COMP. OF 1ST ADDR
7666	6721		TSDF		/LDFRST MUST CONTAIN TWOS COMP. OF NO. WORDS
7667	5266		JMP	,*1	/ TO ACTUALLY STORE
7670	5275		JMP	DBRD	
7671	6722	RDL00P,	TSSR		
7672	5271		JMP	,*1	
7673	6721		TSDF		
7674	5315		JMP	ENDREC	
7675	6715	DBRD,	TSRD		
7676	2337		ISZ	COUNT	
7677	5301		JMP	,*2	
7700	5326		JMP	SZERR	
7701	7430		SZL		/SUPPRESS STORING UNTIL 1ST DESIRED WORD
7702	5307		JMP	STORE	
7703	2354		ISZ	LDFRST	/SKIP WHEN READY TO START STORING
7704	5271		JMP	RDL00P	
7705	7120		STL		
7706	5271		JMP	RDL00P	
7707	3740	STORE,	DCA	I	ADDR /STORE THE WORD READ FROM TAPE
7710	2340		ISZ	ADDR	
7711	2355		ISZ	LDFRST	/SKIP AFTER LAST WORD TO BE STORED IS READ
7712	5271		JMP	RDL00P	
7713	7100		CLL		
7714	5271		JMP	RDL00P	
7715	7200	ENDREC,	CLA		
7716	6734		TSSR		/TAPE STATUS TO AC
7717	7700		SMA	CLA	/SKIP ON BAD PARITY
7720	5323		JMP	ERRCK	
7721	1336		TAD	L2222	
7722	5327		JMP	STOP1	/PARITY ERROR STOP, AC=2222
7723	2337	ERRCK,	ISZ	COUNT	
7724	5326		JMP	,*2	
7725	5662		JMP	I	READ
7726	7240	SZERR,	CLA	CMA	/SIZE ERROR STOP, AC=7777
7727	7402	STOP1,	HLT		/ERROR STOP -- HIT CONTINUE TO TRY AGAIN
7730	5200		JMP	START	
/					
7731	2	C0002,	0002		/COMMAND CODE FOR REWIND
7732	530	C0530,	0530		/COMMAND CODE FOR SPACING
7733	2570	C2570,	2570		/COMMAND CODE FOR READING
7734	7766	0C11,	7766		
7735	176	0C7601,	0176		
7736	2222	L2222,	2222		
7737	0	COUNT,	0000		
7740	0	ADDR,	0000		
7741	7767	M11,	7767		
7742	7776	M2,	7776		
7743	7746	IDL0C,	7746		
/ ONE MORE THAN 1ST ID, LOC. TO MAKE THIS PROG.					
/ READ ADDR1, ADDR2, LDFRST, LDFRST INTO					
/ PROPER PLACES.					
*7745					
7745	0	IDEN0,	0000		/ACTUALLY A DUMMY LOC. -- NOTHING READ INTO HERE
7746	0	ADDR1,	0000		
7747	0	ADDR2,	0000		

7750	0	0000
7751	0	0000
7752	0	0000
7753	0	0000
7754	0	LDFRST, 0000
7755	0	LDLAST, 0000

## SYMBOL TABLE

ADDR	7740
ADDR1	7746
ADDR2	7747
C0002	7731
C0530	7732
C2570	7733
COUNT	7737
DBRD	7675
ENDREC	7715
ERRCK	7723
IDENQ	7745
IDLOC	7743
L2222	7736
LDFRST	7754
LDLAST	7755
M11	7741
M2	7742
OC11	7734
OC7601	7735
RDLAMP	7671
READ	7662
READY	7653
START	7600
STOP1	7727
STORE	7707
SZERR	7726

## DUPLICATE TAGS

NONE

## UNDEFINED SYMBOLS

NONE

Internal Distribution

- |                           |                                       |
|---------------------------|---------------------------------------|
| 1-3. L. S. Abbott         | 21. B. W. Rust                        |
| 4. C. L. Allen (CTC)      | 22. L. W. Weston                      |
| 5. F. E. Bertrand         | 23. G. Dessauer (consultant)          |
| 6. A. A. Brooks (CTC)     | 24. B. C. Diven (consultant)          |
| 7-9. W. R. Burrus         | 25. M. H. Kalos (consultant)          |
| 10. R. E. Funderlic (CTC) | 26. L. V. Spencer (consultant)        |
| 11. T. A. Love            | 27-28. Central Research Library       |
| 12. F. Madden             | 29. Document Reference Section        |
| 13. F. C. Maienschein     | 30-230. Laboratory Records Department |
| 14. E. McDaniel           | 231. Laboratory Records ORNL RC       |
| 15-19. R. W. Peelle       | 232. ORNL Patent Office               |
| 20. F. G. Perey           |                                       |

External Distribution

- 233-247. Division of Technical Information Extension (DTIE)  
248. Division of Research and Development (ORO)